An Unexpected Cyclic Symmetry of Iu$_n$ - Verification Notebook

Pensieve header: Verification notebook for “An Unexpected Cyclic Symmetry of Iu_n by Dror Bar-Natan and Roland van der Veen. Also available at http://drorbn.net/UnexpectedCyclic. Continues pensieve://2020-01/ .

Only Theorem 2 is tested; Theorem 1 is simply the case where $\epsilon = 0$, so it does not require independent testing.

Definitions.

General definitions - brackets *B* and pairings *P* are bilinear, brackets are anti-symmetric:

```
In[•]:= B[0, _] = 0; B[_, 0] = 0;
B[c_ * x : (x | a | b)__, y_] := Expand[c B[x, y]];
B[y_, c_ * x : (x | a | b)__] := Expand[c B[y, x]];
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
```

```
In[•]:= P[0, _] = 0; P[_, 0] = 0;
P[c_ * x : (x | a | b)__, y_] := Expand[c P[x, y]];
P[y_, c_ * x : (x | a | b)__] := Expand[c P[y, x]];
P[x_Plus, y_] := P[#, y] & /@ x;
P[x_, y_Plus] := P[x, #] & /@ y;
```

```
In[•]:= B[y_, x_] := Expand[-B[x, y]];
```

The default value of *n* (can be changed):

```
In[•]:= n = 5;
```

The “length” λ and the “truth indicator” χ_ϵ, and the Kronecker δ-function δ:

```
In[•]:= λ[x_{i_,j_}] := { j - i        i < j
                        { n - (i - j)  i > j
χ_{ϵ_}[cond_] := If[TrueQ@cond, 1, ϵ];
δ_{i_,j_} := χ_0[i == j];
```

The bracket:

```
In[•]:= B[x_{i_,j_}, x_{k_,l_}] := { χ_ϵ[λ[x_{i,j}] + λ[x_{k,l}] < n] (δ_{j,k} x_{i,l} - δ_{l,i} x_{k,j})   j ≠ k ∨ l ≠ i
                                  { b_i - b_j + ϵ/2 a_i - ϵ/2 a_j                                      j == k ∧ l == i ;
B[a_{i_}, x_{j_,k_}] := (δ_{i,j} - δ_{i,k}) x_{j,k};
B[b_{i_}, x_{j_,k_}] := ϵ/2 (δ_{i,j} - δ_{i,k}) x_{j,k};
B[(a | b)_, (a | b)_] = 0;
```

The duality pairing:

In[∘]:=
```
P[x_{i_,j_}, x_{k_,l_}] := δ_{j,k} δ_{l,i};
P[x__, (a | b)_] = 0; P[(a | b)_, x__] = 0;
P[a_{i_}, b_{j_}] := δ_{i,j}; P[b_{j_}, a_{i_}] := δ_{i,j};
P[a_, a_] = 0; P[b_, b_] = 0;
```

The permutation ψ and the automorphism Ψ:

In[∘]:=
```
ψ[k_Integer] := { k + 1   k < n
                { 1       k == n ;
Ψ[ℰ_] := ℰ /. {x_{i_,j_} :→ x_{ψ@i,ψ@j}, a_{i_} :→ a_{ψ@i}, b_{i_} :→ b_{ψ@i}}
```

The basis of $\mathfrak{lu}_n / \mathfrak{gl}^\epsilon_{n+}$:

In[∘]:=
```
Basis[n_] := Flatten@{
    Table[{x_{i,j}, x_{j,i}}, {i, n - 1}, {j, i + 1, n}],
    Table[{a_i, b_i}, {i, n}] }
```

Testing.

In[∘]:=
```
Basis[4]
```

Out[∘]= $\{x_{1,2}, x_{2,1}, x_{1,3}, x_{3,1}, x_{1,4}, x_{4,1}, x_{2,3}, x_{3,2}, x_{2,4}, x_{4,2}, x_{3,4}, x_{4,3}, a_1, b_1, a_2, b_2, a_3, b_3, a_4, b_4\}$

A full bracket-table for $n = 2$:

In[∘]:=
```
n = 2; MatrixForm[
 Table[B[u, v], {u, Basis[n]}, {v, Basis[n]}],
 TableHeadings → {Basis[n], Basis[n]}]
```

Out[∘]//MatrixForm=

	$x_{1,2}$	$x_{2,1}$	a_1	b_1	a_2	b_2
$x_{1,2}$	0	$\frac{\epsilon a_1}{2} - \frac{\epsilon a_2}{2} + b_1 - b_2$	$-x_{1,2}$	$-\frac{1}{2}\epsilon x_{1,2}$	$x_{1,2}$	$\frac{1}{2}\epsilon x_{1,2}$
$x_{2,1}$	$-\frac{\epsilon a_1}{2} + \frac{\epsilon a_2}{2} - b_1 + b_2$	0	$x_{2,1}$	$\frac{1}{2}\epsilon x_{2,1}$	$-x_{2,1}$	$-\frac{1}{2}\epsilon x_{2,1}$
a_1	$x_{1,2}$	$-x_{2,1}$	0	0	0	0
b_1	$\frac{1}{2}\epsilon x_{1,2}$	$-\frac{1}{2}\epsilon x_{2,1}$	0	0	0	0
a_2	$-x_{1,2}$	$x_{2,1}$	0	0	0	0
b_2	$-\frac{1}{2}\epsilon x_{1,2}$	$\frac{1}{2}\epsilon x_{2,1}$	0	0	0	0

The bracket is anti-symmetric at $n = 4$:

In[∘]:=
```
n = 4; Short@Table[
  {u, v} = t; B[u, v] + B[v, u],
  {t, Tuples[Basis[n], 2]}]
```

Out[∘]//Short=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,
 0, <<364>>, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

The bracket satisfies the Jacobi identity:

In[∘]:=
```
n = 4; DeleteCases[0]@Table[
  {u, v, w} = t; B[u, B[v, w]] + B[v, B[w, u]] + B[w, B[u, v]],
  {t, Tuples[Basis[n], 3]} ]
```

Out[∘]= { }

The pairing is invariant:

```
In[∘]:= n = 4; DeleteCases[0]@Table[
          {u, v, w} = t; P[B[u, v], w] + P[v, B[u, w]],
          {t, Tuples[Basis[n], 3]} ]
Out[∘]= { }
```

The action of Ψ:

```
In[∘]:= (# → Ψ[#]) & /@ Basis[4]
```

Out[∘]= $\{x_{1,2} \to x_{2,3}, x_{2,1} \to x_{3,2}, x_{1,3} \to x_{2,4}, x_{3,1} \to x_{4,2}, x_{1,4} \to x_{2,1}, x_{4,1} \to x_{1,2}, x_{2,3} \to x_{3,4}, x_{3,2} \to x_{4,3}, x_{2,4} \to x_{3,1}, x_{4,2} \to x_{1,3}, x_{3,4} \to x_{4,1}, x_{4,3} \to x_{1,4}, a_1 \to a_2, b_1 \to b_2, a_2 \to a_3, b_2 \to b_3, a_3 \to a_4, b_3 \to b_4, a_4 \to a_1, b_4 \to b_1\}$

Ψ is an automorphism:

```
In[∘]:= n = 4; DeleteCases[0]@Table[
          {u, v} = t; Ψ[B[u, v]] - B[Ψ[u], Ψ[v]],
          {t, Tuples[Basis[n], 2]} ]
Out[∘]= { }
```

Ψ respects the pairing:

```
In[∘]:= n = 4; DeleteCases[0]@Table[
          {u, v} = t; Ψ[P[u, v]] - P[Ψ[u], Ψ[v]],
          {t, Tuples[Basis[n], 2]} ]
Out[∘]= { }
```

Bonus Tests

Acting by arbitrary index-permutations:

```
In[∘]:= Act_σ_List[ℰ_] := ℰ /. {x_{i_,j_} :> x_{σ[[i]],σ[[j]]}, a_{i_} :> a_{σ[[i]]}, b_{i_} :> b_{σ[[i]]}}
```

At $n = 5$, only cyclic permutations induce automorphisms:

```
In[∘]:= n = 5;
        Select[Permutations[Range[n]],
          σ ↦ And @@ Flatten[Table[
               Act_σ[B[u, v]] === B[Act_σ[u], Act_σ[v]],
               {u, Basis[n]}, {v, Basis[n]}
             ]]
        ]
Out[∘]= {{1, 2, 3, 4, 5}, {2, 3, 4, 5, 1}, {3, 4, 5, 1, 2}, {4, 5, 1, 2, 3}, {5, 1, 2, 3, 4}}
```

Yet in the case of gl_n, meaning when $\epsilon = 1$, all permutations induce automorphisms:

In[∘]:=
```
n = 4;
Block[{ϵ = 1}, Select[Permutations[Range[n]],
   σ ↦ And @@ Flatten[Table[
       Act_σ[B[u, v]] === B[Act_σ[u], Act_σ[v]],
       {u, Basis[n]}, {v, Basis[n]}
      ]]
 ]]
```

Out[∘]=
```
{{1, 2, 3, 4}, {1, 2, 4, 3}, {1, 3, 2, 4}, {1, 3, 4, 2}, {1, 4, 2, 3}, {1, 4, 3, 2},
 {2, 1, 3, 4}, {2, 1, 4, 3}, {2, 3, 1, 4}, {2, 3, 4, 1}, {2, 4, 1, 3}, {2, 4, 3, 1},
 {3, 1, 2, 4}, {3, 1, 4, 2}, {3, 2, 1, 4}, {3, 2, 4, 1}, {3, 4, 1, 2}, {3, 4, 2, 1},
 {4, 1, 2, 3}, {4, 1, 3, 2}, {4, 2, 1, 3}, {4, 2, 3, 1}, {4, 3, 1, 2}, {4, 3, 2, 1}}
```

If ϵ is invertible, the isomorphism class of gl_{n+}^{ϵ} is independent of ϵ, using Inonu-Wigner contractions:

In[∘]:=
```
IW_λ_[ℰ_] := ℰ /. {x_{i_,j_} /; i > j :→ λ x_{i,j}, b_{i_} :→ λ b_i};
```

In[∘]:=
```
n = 4;
Union@Flatten@Table[
   (B[u, v] /. ϵ → 1) == IW_ϵ@B[IW_{1/ϵ}@u, IW_{1/ϵ}@v],
   {u, Basis[n]}, {v, Basis[n]}
  ]
```

Out[∘]=
```
{True}
```

Even cyclic index permutations become singular at $\epsilon = 0$ when conjugated by Inonu-Wigner contractions (so Ψ simply isn't that):

In[∘]:=
```
n = 4; Table[u → IW_ϵ@Act_{2,3,4,1}@IW_{1/ϵ}@u, {u, Basis[n]}]
```

Out[∘]=

$$\left\{x_{1,2}\to x_{2,3},\ x_{2,1}\to x_{3,2},\ x_{1,3}\to x_{2,4},\ x_{3,1}\to x_{4,2},\ x_{1,4}\to \epsilon\, x_{2,1},\ x_{4,1}\to \frac{x_{1,2}}{\epsilon},\ x_{2,3}\to x_{3,4},\ x_{3,2}\to x_{4,3},\ x_{2,4}\to \epsilon\, x_{3,1},\ x_{4,2}\to \frac{x_{1,3}}{\epsilon},\ x_{3,4}\to \epsilon\, x_{4,1},\ x_{4,3}\to \frac{x_{1,4}}{\epsilon},\ a_1\to a_2,\ b_1\to b_2,\ a_2\to a_3,\ b_2\to b_3,\ a_3\to a_4,\ b_3\to b_4,\ a_4\to a_1,\ b_4\to b_1\right\}$$

In[∘]:=
```
n = 4; Table[u → IW_{1/ϵ}@Act_{2,3,4,1}@IW_ϵ@u, {u, Basis[n]}]
```

Out[∘]=

$$\left\{x_{1,2}\to x_{2,3},\ x_{2,1}\to x_{3,2},\ x_{1,3}\to x_{2,4},\ x_{3,1}\to x_{4,2},\ x_{1,4}\to \frac{x_{2,1}}{\epsilon},\ x_{4,1}\to \epsilon\, x_{1,2},\ x_{2,3}\to x_{3,4},\ x_{3,2}\to x_{4,3},\ x_{2,4}\to \frac{x_{3,1}}{\epsilon},\ x_{4,2}\to \epsilon\, x_{1,3},\ x_{3,4}\to \frac{x_{4,1}}{\epsilon},\ x_{4,3}\to \epsilon\, x_{1,4},\ a_1\to a_2,\ b_1\to b_2,\ a_2\to a_3,\ b_2\to b_3,\ a_3\to a_4,\ b_3\to b_4,\ a_4\to a_1,\ b_4\to b_1\right\}$$

The same is true for all other permutations (except the identity):

In[∘]:=
```
n = 3; MatrixForm[
  Table[IW_ϵ@Act_σ@IW_{1/ϵ}@u, {σ, rows = Permutations@Range@n}, {u, cols = Basis[n]}],
  TableHeadings → {rows, cols} ]
```

Out[∘]//MatrixForm=

	$x_{1,2}$	$x_{2,1}$	$x_{1,3}$	$x_{3,1}$	$x_{2,3}$	$x_{3,2}$	a_1	b_1	a_2	b_2	a_3	b_3
{1, 2, 3}	$x_{1,2}$	$x_{2,1}$	$x_{1,3}$	$x_{3,1}$	$x_{2,3}$	$x_{3,2}$	a_1	b_1	a_2	b_2	a_3	b_3
{1, 3, 2}	$x_{1,3}$	$x_{3,1}$	$x_{1,2}$	$x_{2,1}$	$\epsilon\, x_{3,2}$	$\frac{x_{2,3}}{\epsilon}$	a_1	b_1	a_3	b_3	a_2	b_2
{2, 1, 3}	$\epsilon\, x_{2,1}$	$\frac{x_{1,2}}{\epsilon}$	$x_{2,3}$	$x_{3,2}$	$x_{1,3}$	$x_{3,1}$	a_2	b_2	a_1	b_1	a_3	b_3
{2, 3, 1}	$x_{2,3}$	$x_{3,2}$	$\epsilon\, x_{2,1}$	$\frac{x_{1,2}}{\epsilon}$	$\epsilon\, x_{3,1}$	$\frac{x_{1,3}}{\epsilon}$	a_2	b_2	a_3	b_3	a_1	b_1
{3, 1, 2}	$\epsilon\, x_{3,1}$	$\frac{x_{1,3}}{\epsilon}$	$\epsilon\, x_{3,2}$	$\frac{x_{2,3}}{\epsilon}$	$x_{1,2}$	$x_{2,1}$	a_3	b_3	a_1	b_1	a_2	b_2
{3, 2, 1}	$\epsilon\, x_{3,2}$	$\frac{x_{2,3}}{\epsilon}$	$\epsilon\, x_{3,1}$	$\frac{x_{1,3}}{\epsilon}$	$\epsilon\, x_{2,1}$	$\frac{x_{1,2}}{\epsilon}$	a_3	b_3	a_2	b_2	a_1	b_1

In[∘]:=
```
n = 3; MatrixForm[
  Table[IW_{1/ϵ}@Act_σ@IW_ϵ@u, {σ, rows = Permutations@Range@n}, {u, cols = Basis[n]}],
  TableHeadings → {rows, cols} ]
```

Out[∘]//MatrixForm=

	$x_{1,2}$	$x_{2,1}$	$x_{1,3}$	$x_{3,1}$	$x_{2,3}$	$x_{3,2}$	a_1	b_1	a_2	b_2	a_3	b_3
{1, 2, 3}	$x_{1,2}$	$x_{2,1}$	$x_{1,3}$	$x_{3,1}$	$x_{2,3}$	$x_{3,2}$	a_1	b_1	a_2	b_2	a_3	b_3
{1, 3, 2}	$x_{1,3}$	$x_{3,1}$	$x_{1,2}$	$x_{2,1}$	$\frac{x_{3,2}}{\epsilon}$	$\epsilon\, x_{2,3}$	a_1	b_1	a_3	b_3	a_2	b_2
{2, 1, 3}	$\frac{x_{2,1}}{\epsilon}$	$\epsilon\, x_{1,2}$	$x_{2,3}$	$x_{3,2}$	$x_{1,3}$	$x_{3,1}$	a_2	b_2	a_1	b_1	a_3	b_3
{2, 3, 1}	$x_{2,3}$	$x_{3,2}$	$\frac{x_{2,1}}{\epsilon}$	$\epsilon\, x_{1,2}$	$\frac{x_{3,1}}{\epsilon}$	$\epsilon\, x_{1,3}$	a_2	b_2	a_3	b_3	a_1	b_1
{3, 1, 2}	$\frac{x_{3,1}}{\epsilon}$	$\epsilon\, x_{1,3}$	$\frac{x_{3,2}}{\epsilon}$	$\epsilon\, x_{2,3}$	$x_{1,2}$	$x_{2,1}$	a_3	b_3	a_1	b_1	a_2	b_2
{3, 2, 1}	$\frac{x_{3,2}}{\epsilon}$	$\epsilon\, x_{2,3}$	$\frac{x_{3,1}}{\epsilon}$	$\epsilon\, x_{1,3}$	$\frac{x_{2,1}}{\epsilon}$	$\epsilon\, x_{1,2}$	a_3	b_3	a_2	b_2	a_1	b_1